
************************************************************************************************************************************************
*    Results prison conditions and recidivism ****
*    Version 6
*    November 25 2020
************************************************************************************************************************************************

clear all

	set matsize 4000
	use database_prison_conditions, clear
	
	
* Control vectors:

	global cvt = "new_offender cri_d_violent_any d_homicide d_assault d_sex_crime d_other_violent cri_d_nonviolent_any d_property d_conspiracy d_other_crime cri_d_drug_any age_entry ed_none ed_primary ed_secondary ed_tertiary minors y_ingreso"

	
* Final adjustments to outcomes and variables for analysis over time

	forvalues s = 1(1)36  {
				
		gen g3_`s' = g3
			
	} 
	
* Sequencing for tests
	
	* Sequences for entry to any prison
		
		sort dj_establ_id f_ingreso con_cap
		
		gen rep = (cod_establ_aj == cod_establ_aj[_n-1])
		replace rep = . if con_cap[_n+1] == 1

		gen seq = 0 if rep == 0
		replace seq = rep + seq[_n-1] if rep == 1
		replace seq = seq + 1
		replace seq = . if rep == 1 & rep[_n+1] == 1
		replace seq = . if rep[_n+1] == 1
		
		gen no_rep = !rep // outcome (change of prison)

		tab dj_establ_id g3, su(seq)
		
		
* Descriptive statistics

	
	* Table 2 (summary statistics part)
	
		eststo clear
		eststo: quietly estpost sum new_offender cri_d_violent_any d_homicide d_assault d_sex_crime d_other_violent cri_d_nonviolent_any d_property d_conspiracy d_other_crime cri_d_drug_any age_entry ed_none ed_primary ed_secondary ed_tertiary minors y_ingreso g3 sec_age_release dias_interno re_entry_1y re_conv_1y net_size new_entries reh_trabajo reh_educ baja_muerte visits if valid_events & jud_convicted & re_entry_1y != .
		esttab using "revision/main_summary.csv", cells("mean(fmt(%9.3f)) sd(fmt(%9.3f)) min(fmt(%9.0f)) max(fmt(%9.0f))") label nodepvar replace

		
* Identification

	* Table 3 (summary statistics for sequence size part)
	
		levelsof dj_establ_id, local(levels)
		
		putexcel set "revision/distritos.xlsx", replace
		putexcel A1=("District") B1=("No. of prisons") C1=("Total cases") D1=("Cases in analytical sample") E1=("No. of sequences") F1=("Mean") G1=("S.D.") H1=("Min") I1=("Max")
		local s 1
	
		foreach l of local levels {
		
			local ++s
		
			di "`l'"
			putexcel A`s'=("`l'")
			
			distinct cod_establ_aj if dj_establ_id == `l'
			putexcel B`s'=(r(ndistinct))
			
			distinct evento if dj_establ_id == `l'
			putexcel C`s'=(r(ndistinct))
			
			distinct evento if dj_establ_id == `l' & jud_convicted & re_entry_1y != .
			putexcel D`s'=(r(ndistinct))

			su seq if dj_establ_id == `l', d
			putexcel E`s'=(r(N))
			putexcel F`s'=(r(mean))
			putexcel G`s'=(r(sd))
			putexcel H`s'=(r(min))
			putexcel I`s'=(r(max))
			
		}
		
		local ++s
		
		distinct cod_establ_aj
		putexcel B`s'=(r(ndistinct))
		
		distinct evento
		putexcel C`s'=(r(ndistinct))
		
		distinct evento if jud_convicted & re_entry_1y != .
		putexcel D`s'=(r(ndistinct))

		su seq, d
		putexcel E`s'=(r(N))
		putexcel F`s'=(r(mean))
		putexcel G`s'=(r(sd))
		putexcel H`s'=(r(min))
		putexcel I`s'=(r(max))
		
		putexcel set "revision/distritos.xlsx", replace
		
		
		* Table B2 (summary statistics for sequence size part)
		
			levelsof dj_establ_id, local(levels)
		
			putexcel set "revision/distritos_new.xlsx", replace
			putexcel A1=("District") B1=("No. of prisons") C1=("Total cases") D1=("Cases in analytical sample") E1=("No. of sequences") F1=("Mean") G1=("S.D.") H1=("Min") I1=("Max")
			local s 1
		
			foreach l of local levels {
			
				local ++s
			
				di "`l'"
				putexcel A`s'=("`l'")
				
				distinct cod_establ_aj if dj_establ_id == `l' & g3
				putexcel B`s'=(r(ndistinct))
				
				distinct evento if dj_establ_id == `l' & g3
				putexcel C`s'=(r(ndistinct))
				
				distinct evento if dj_establ_id == `l' & jud_convicted & re_entry_1y != . & g3
				putexcel D`s'=(r(ndistinct))

				su seq if dj_establ_id == `l' & g3, d
				putexcel E`s'=(r(N))
				putexcel F`s'=(r(mean))
				putexcel G`s'=(r(sd))
				putexcel H`s'=(r(min))
				putexcel I`s'=(r(max))
				
			}
			
			local ++s
			
			distinct cod_establ_aj if g3
			putexcel B`s'=(r(ndistinct))
			
			distinct evento if g3
			putexcel C`s'=(r(ndistinct))
			
			distinct evento if jud_convicted & re_entry_1y != . & g3
			putexcel D`s'=(r(ndistinct))

			su seq if g3, d
			putexcel E`s'=(r(N))
			putexcel F`s'=(r(mean))
			putexcel G`s'=(r(sd))
			putexcel H`s'=(r(min))
			putexcel I`s'=(r(max))
			
			putexcel set "revision/distritos_new.xlsx", replace

			
		* Table B3 (summary statistics for sequence size part)
		
			levelsof dj_establ_id, local(levels)
		
			putexcel set "revision/distritos_old.xlsx", replace
			putexcel A1=("District") B1=("No. of prisons") C1=("Total cases") D1=("Cases in analytical sample") E1=("No. of sequences") F1=("Mean") G1=("S.D.") H1=("Min") I1=("Max")
			local s 1
		
			foreach l of local levels {
			
				local ++s
			
				di "`l'"
				putexcel A`s'=("`l'")
				
				distinct cod_establ_aj if dj_establ_id == `l' & !g3
				putexcel B`s'=(r(ndistinct))
				
				distinct evento if dj_establ_id == `l' & !g3
				putexcel C`s'=(r(ndistinct))
				
				distinct evento if dj_establ_id == `l' & jud_convicted & re_entry_1y != . & !g3
				putexcel D`s'=(r(ndistinct))

				su seq if dj_establ_id == `l' & !g3, d
				putexcel E`s'=(r(N))
				putexcel F`s'=(r(mean))
				putexcel G`s'=(r(sd))
				putexcel H`s'=(r(min))
				putexcel I`s'=(r(max))
				
			}
			
			local ++s
			
			distinct cod_establ_aj if !g3
			putexcel B`s'=(r(ndistinct))
			
			distinct evento if !g3
			putexcel C`s'=(r(ndistinct))
			
			distinct evento if jud_convicted & re_entry_1y != . & !g3
			putexcel D`s'=(r(ndistinct))

			su seq if !g3, d
			putexcel E`s'=(r(N))
			putexcel F`s'=(r(mean))
			putexcel G`s'=(r(sd))
			putexcel H`s'=(r(min))
			putexcel I`s'=(r(max))
			
			putexcel set "revision/distritos_old.xlsx", replace

	
		* Table 2 (baseline balance test on prison assignment)
			
			putexcel set "revision/balance_2nd_rev.xlsx", replace
			putexcel A1=("Variable") B1=("Coeff.") C1=("S.E.")
			local s 1
		
			foreach c in $cvt {
				
				local ++s
		
				qui reg g3 `c' _W* _X* if jud_convicted & re_entry_1y != ., robust cluster(cod_establ_aj)
				
				di "`c'"
				di _b[`c']
				test _b[`c'] = 0
				
				putexcel A`s'=("`c'") 
				putexcel B`s'=(_b[`c'])
				putexcel C`s'=(_se[`c'])
		
			}
			putexcel set "revision/balance_2nd_rev.xlsx", replace
			
			
		* Table 2 (baseline balance test on sequence breaks)
			
			
			putexcel set "revision/balance_no_rep_2nd_rev.xlsx", replace
			putexcel A1=("Variable") B1=("Coeff.") C1=("S.E.")
			local s 1
		
			foreach c in $cvt {
				
				local ++s
		
				qui reg no_rep `c' _W* _X* if jud_convicted & re_entry_1y != ., robust cluster(cod_establ_aj)
				
				di "`c'"
				di _b[`c']
				test _b[`c'] = 0
				
				putexcel A`s'=("`c'") 
				putexcel B`s'=(_b[`c'])
				putexcel C`s'=(_se[`c'])
		
			}
			putexcel set "revision/balance_no_rep_2nd_rev.xlsx", replace
	

	* Table 2 (p value of F tests of joint significance)
	
		eststo clear
		
			reg g3 new_offender cri_* sec_* dias_interno _W* _X* if jud_convicted & re_entry_1y != ., robust cluster(cod_establ_aj)
			estimates store m1, title(Model 1)
			test (new_offender=0) (cri_d_violent_any=0) (cri_d_nonviolent_any=0) (cri_d_drug_any=0) (sec_age_release=0) (sec_age_release_2=0) (dias_interno=0)
			
			
			reg no_rep new_offender cri_* sec_* dias_interno _W* _X* if jud_convicted & re_entry_1y != ., robust cluster(cod_establ_aj)
			estimates store m2, title(Model 2)
			test (new_offender=0) (cri_d_violent_any=0) (cri_d_nonviolent_any=0) (cri_d_drug_any=0) (sec_age_release=0) (sec_age_release_2=0) (dias_interno=0)
		
		estout m1 m2 using "revision/preinc.csv", replace style(fixed) cells(b(star fmt(3)) se(par([ ]) fmt(3)))  ///
			legend nolabel starlevels( * 0.10 ** 0.05 *** 0.010) drop(_D* _cons)             ///
			stats(r2 N, fmt(3 0) label(R-2 Obs.))
		
			
		
		
* Results
	
	* Table 4 (Panel A)
	
		eststo clear
	
		reg re_entry_1y g3 sec_* if jud_convicted & re_entry_1y != ., robust cluster(cod_establ_aj)
		estimates store m1, title(Model 1)
		
		reg re_entry_1y g3 new_offender cri_* sec_* if jud_convicted & re_entry_1y != ., robust cluster(cod_establ_aj)
		estimates store m2, title(Model 2)
		
		reg re_entry_1y g3 new_offender cri_* sec_* dias_interno if jud_convicted & re_entry_1y != ., robust cluster(cod_establ_aj)
		estimates store m3, title(Model 3)
		
		reg re_entry_1y g3 new_offender cri_* sec_* dias_interno _W* if jud_convicted & re_entry_1y != ., robust cluster(cod_establ_aj)
		estimates store m4, title(Model 4)
		
		reg re_entry_1y g3 new_offender cri_* sec_* dias_interno _W* _X* if jud_convicted & re_entry_1y != ., robust cluster(cod_establ_aj)
		estimates store m5, title(Model 5)
		
		estout m1 m2 m3 m4 m5 using "revision/baseline.csv", replace style(fixed) cells(b(star fmt(3)) se(par([ ]) fmt(3)))  ///
			legend nolabel starlevels( * 0.10 ** 0.05 *** 0.010) drop(new_offender cri_* sec_* dias_interno _W* _X* _cons)     ///
			stats(r2 N, fmt(3 0) label(R-2 Obs.))
			
		
		
		* Table 2 (additional test with summary measure of risk)
	
			reg re_entry_1y new_offender cri_* sec_* dias_interno _W* _X* if jud_convicted & re_entry_1y != ., robust cluster(cod_establ_aj)
			predict p_re_entry_1y if e(sample) 
			
			reg g3 p_re_entry_1y _W* _X* if jud_convicted & re_entry_1y != ., robust cluster(cod_establ_aj)
			
			reg no_rep p_re_entry_1y _W* _X* if jud_convicted & re_entry_1y != ., robust cluster(cod_establ_aj)
			
	
		
		* Figure 3
					
			local s 0
			local mo 0

			foreach k in 31 28 31 30 31 30 31 31 30 31 30 31 31 28 31 30 31 30 31 31 30 31 30 31 31 28 31 30 31 30 31 31 30 31 30 31 {
			
				local ++s
				local mo = `mo' + `k'

				qui reg re_entry_t`s' g3_`s' new_offender cri_* sec_* dias_interno _W* _X* if jud_convicted & re_entry_t`s' != ., robust cluster(cod_establ_aj)
				estimates store t`s'
				di `s'
				di _b[g3_`s']
				di `e(N)'
				qui su re_entry_t`s' if jud_convicted & re_entry_t`s' != . & !g3_`s'
				di r(mean)

			}
			
			graph set window fontface "Times New Roman"	
			
			coefplot 	(t1, ciopts(lcolor(black))) (t2, ciopts(lcolor(black))) (t3, ciopts(lcolor(black))) (t4, ciopts(lcolor(black))) (t5, ciopts(lcolor(black))) (t6, ciopts(lcolor(black))) (t7, ciopts(lcolor(black))) (t8, ciopts(lcolor(black))) (t9, ciopts(lcolor(black))) (t10, ciopts(lcolor(black))) (t11, ciopts(lcolor(black))) (t12, ciopts(lcolor(black))) (t13, ciopts(lcolor(black))) (t14, ciopts(lcolor(black))) (t15, ciopts(lcolor(black))) (t16, ciopts(lcolor(black))) (t17, ciopts(lcolor(black))) (t18, ciopts(lcolor(black))) (t19, ciopts(lcolor(black))) (t20, ciopts(lcolor(black))) (t21, ciopts(lcolor(black))) (t22, ciopts(lcolor(black))) (t23, ciopts(lcolor(black))) (t24, ciopts(lcolor(black))) (t25, ciopts(lcolor(black))) (t26, ciopts(lcolor(black))) (t27, ciopts(lcolor(black))) (t28, ciopts(lcolor(black))) (t29, ciopts(lcolor(black))) (t30, ciopts(lcolor(black))) (t31, ciopts(lcolor(black))) (t32, ciopts(lcolor(black))) (t33, ciopts(lcolor(black))) (t34, ciopts(lcolor(black))) (t35, ciopts(lcolor(black))) (t36, ciopts(lcolor(black))), ///
						coeflabels(g3_1 = "1" g3_2 = "2" g3_3 = "3" g3_4 = "4" g3_5 = "5" g3_6 = "6" g3_7 = "7" g3_8 = "8" g3_9 = "9" g3_10 = "10" g3_11 = "11" g3_12 = "12" g3_13 = "13" g3_14 = "14" g3_15 = "15" g3_16 = "16" g3_17 = "17" g3_18 = "18" g3_19 = "19" g3_20 = "20" g3_21 = "21" g3_22 = "22" g3_23 = "23" g3_24 = "24" g3_25 = "25" g3_26 = "26" g3_27 = "27" g3_28 = "28" g3_29 = "29" g3_30 = "30" g3_31 = "31" g3_32 = "32" g3_33 = "33" g3_34 = "34" g3_35 = "35" g3_36 = "36")  ///
						vertical drop(new_offender cri_* sec_* dias_interno _W* _X* _cons) ///
						yline(0, lcolor(black)) ytitle(Point estimate) xtitle(Months since release) xlabel(, angle(vertical)) yscale(r(-.16 .04)) ylabel(-.16(.04).04, angle(vertical)) graphregion(fcolor(white)) levels(95) legend(off) msymbol(D) mlcolor(black) mfcolor(black) ciopts(lwidth(*1)) name(b4, replace)
			
			graph export "robustness_OLS_alone36.png",  replace width(2000)
			window manage close graph


	*Mechanisms
	
		* Table 5 (Panel A - criminal capital)
		
			eststo clear
		
			qui reg net_size g3 new_offender cri_* sec_* dias_interno _W* _X* if jud_convicted & re_entry_1y != ., robust cluster(cod_establ_aj)
			estimates store m1, title(Model 1)
			
			su net_size if !g3 & jud_convicted & re_entry_1y != .
			
			qui reg new_entries g3 new_offender cri_* sec_* dias_interno _W* _X* if jud_convicted & re_entry_1y != ., robust cluster(cod_establ_aj)
			estimates store m2, title(Model 2)
			
			su new_entries if !g3 & jud_convicted & re_entry_1y != .
			
			
			
			*Sub-sample analysis for new offenders
			
				* Re entry 1y
			
					qui reg re_entry_1y g3 new_offender cri_* sec_* dias_interno _W* _X* if jud_convicted & re_entry_1y != . & new_offender, robust cluster(cod_establ_aj)
					estimates store m3, title(Model 3)
					
					su re_entry_1y if !g3 & jud_convicted & re_entry_1y != . & new_offender
					
					qui reg re_entry_1y g3 new_offender cri_* sec_* dias_interno _W* _X* if jud_convicted & re_entry_1y != . & !new_offender, robust cluster(cod_establ_aj)
					estimates store m4, title(Model 4)
					
					su re_entry_1y if !g3 & jud_convicted & re_entry_1y != . & !new_offender
					
					* Table B4 (Panel A)
					
						reg re_entry_1y g3 new_offender g3_x_new_offender cri_* sec_* dias_interno _W* _X* if jud_convicted & re_entry_1y != ., robust cluster(cod_establ_aj)
				
				* New entries
				
					qui reg new_entries g3 new_offender cri_* sec_* dias_interno _W* _X* if jud_convicted & re_entry_1y != . & new_offender, robust cluster(cod_establ_aj)
					estimates store m5, title(Model 5)
					
					su new_entries if !g3 & jud_convicted & re_entry_1y != . & new_offender
					
					qui reg new_entries g3 new_offender cri_* sec_* dias_interno _W* _X* if jud_convicted & re_entry_1y != . & !new_offender, robust cluster(cod_establ_aj)
					estimates store m6, title(Model 6)
					
					su new_entries if !g3 & jud_convicted & re_entry_1y != . & !new_offender
					
					* Table B4 (Panel B)
					
						reg new_entries g3 new_offender g3_x_new_offender cri_* sec_* dias_interno _W* _X* if jud_convicted & re_entry_1y != ., robust cluster(cod_establ_aj)
						

		* Table 5 (Panel B - rehabilitation)
					
			qui reg reh_trabajo g3 new_offender cri_* sec_* dias_interno _W* _X* if jud_convicted & re_entry_1y != ., robust cluster(cod_establ_aj)
			estimates store m7, title(Model 7)
			
			su reh_trabajo if !g3 & jud_convicted & re_entry_1y != . 
			
			qui reg reh_educ g3 new_offender cri_* sec_* dias_interno _W* _X* if jud_convicted & re_entry_1y != ., robust cluster(cod_establ_aj)
			estimates store m8, title(Model 8)
			
			su reh_educ if !g3 & jud_convicted & re_entry_1y != . 
			
			
			* Sub-sample analysis for rehabilitation
								
						
				* Rehab working
				
					qui reg reh_trabajo g3 new_offender cri_* sec_* dias_interno _W* _X* if jud_convicted & re_entry_1y != . & ed_none, robust cluster(cod_establ_aj)
					estimates store m13, title(Model 9)
					
					su reh_trabajo if !g3 & jud_convicted & re_entry_1y != . & ed_none
					
					qui reg reh_trabajo g3 new_offender cri_* sec_* dias_interno _W* _X* if jud_convicted & re_entry_1y != . & ed_primary, robust cluster(cod_establ_aj)
					estimates store m14, title(Model 10)
					
					su reh_trabajo if !g3 & jud_convicted & re_entry_1y != . & ed_primary
					
					qui reg reh_trabajo g3 new_offender cri_* sec_* dias_interno _W* _X* if jud_convicted & re_entry_1y != . & ed_secondary, robust cluster(cod_establ_aj)
					estimates store m15, title(Model 11)
					
					su reh_trabajo if !g3 & jud_convicted & re_entry_1y != . & ed_secondary
					
					qui reg reh_trabajo g3 new_offender cri_* sec_* dias_interno _W* _X* if jud_convicted & re_entry_1y != . & ed_tertiary, robust cluster(cod_establ_aj) //aquí se concentra este efecto
					estimates store m16, title(Model 12)
					
					su reh_trabajo if !g3 & jud_convicted & re_entry_1y != . & ed_tertiary
								
					
				* Rehab education
					
					qui reg reh_educ g3 new_offender cri_* sec_* dias_interno _W* _X* if jud_convicted & re_entry_1y != . & ed_none, robust cluster(cod_establ_aj)
					estimates store m17, title(Model 13)
					
					su reh_educ if !g3 & jud_convicted & re_entry_1y != . & ed_none
					
					qui reg reh_educ g3 new_offender cri_* sec_* dias_interno _W* _X* if jud_convicted & re_entry_1y != . & ed_primary, robust cluster(cod_establ_aj)
					estimates store m18, title(Model 14)
					
					su reh_educ if !g3 & jud_convicted & re_entry_1y != . & ed_primary
					
					qui reg reh_educ g3 new_offender cri_* sec_* dias_interno _W* _X* if jud_convicted & re_entry_1y != . & ed_secondary, robust cluster(cod_establ_aj)
					estimates store m19, title(Model 15)
					
					su reh_educ if !g3 & jud_convicted & re_entry_1y != . & ed_secondary
					
					qui reg reh_educ g3 new_offender cri_* sec_* dias_interno _W* _X* if jud_convicted & re_entry_1y != . & ed_tertiary, robust cluster(cod_establ_aj)
					estimates store m20, title(Model 16)
					
					su reh_educ if !g3 & jud_convicted & re_entry_1y != . & ed_tertiary	
					
		* Table 5 (Panel C - prison experience)
			
			qui reg baja_muerte g3 new_offender cri_* sec_* dias_interno _W* _X* if jud_convicted & re_entry_1y != ., robust cluster(cod_establ_aj)
			estimates store m21, title(Model 17)
			
			su baja_muerte if !g3 & jud_convicted & re_entry_1y != .
			
			qui reg visits g3 new_offender cri_* sec_* dias_interno _W* _X* if jud_convicted & re_entry_1y != ., robust cluster(cod_establ_aj)
			estimates store m22, title(Model 18)
			
			su visits if !g3 & jud_convicted & re_entry_1y != .
			
			estout m1 m2 m3 m4 m5 m6 m7 m8 m9 m10 m11 m12 m13 m14 m15 m16 m17 m18 using "revision/mechanisms.csv", replace style(fixed) cells(b(star fmt(3)) se(par([ ]) fmt(3)))  ///
				legend nolabel starlevels( * 0.10 ** 0.05 *** 0.010) drop(new_offender cri_* sec_* dias_interno _W* _X* _cons)     ///
				stats(r2 N, fmt(3 0) label(R-2 Obs.))
			
			
			
	* Table 4 (Panel B)

		eststo clear
			
		qui reg re_entry_1y g3 new_offender cri_* sec_* dias_interno ed_primary ed_secondary ed_tertiary minors _W* _X* if jud_convicted & re_entry_1y != ., robust cluster(cod_establ_aj)
		estimates store m1, title(Model 1)
		
		qui reg re_entry_1y g3 new_offender d_homicide d_assault d_sex_crime d_other_violent d_property d_conspiracy d_other_crime cri_d_drug_any sec_* dias_interno _W* _X* if jud_convicted & re_entry_1y != ., robust cluster(cod_establ_aj)
		estimates store m2, title(Model 2)
		
		qui reg re_conv_1y g3 new_offender cri_* sec_* dias_interno _W* _X* if jud_convicted & re_entry_1y != ., robust cluster(cod_establ_aj)
		estimates store m3, title(Model 3)
		su re_conv_1y if !g3 & jud_convicted & re_entry_1y != .
		
		qui reg re_entry_1y g3 new_offender cri_* sec_* dias_interno _W* _X* if re_entry_1y != ., robust cluster(cod_establ_aj)
		estimates store m4, title(Model 4)
		su re_entry_1y if !g3 & re_entry_1y != .
		
		su dias_interno if re_entry_1y != .
		su dias_interno if re_entry_1y != . & !jud_convicted
		
		levelsof dj_establ_id, local(levels) 
		local s 4
	
		foreach l of local levels {
			
			local ++s
		
			qui reg re_entry_1y g3 new_offender cri_* sec_* dias_interno _W* _X* if jud_convicted & re_entry_1y != . & dj_establ_id != `l', robust cluster(cod_establ_aj)
			estimates store m`s', title(Model `s')
			
			di "`l'"
			di _b[g3]
			test _b[g3] = 0
			su re_entry_1y if !g3 & jud_convicted & re_entry_1y != . & dj_establ_id != `l'
			
		}
		
		
		
		estout m1 m2 m3 m4 m5 m6 m7 m8 m9 m10 m11 m12 m13 m14 using "revision/robustness.csv", replace style(fixed) cells(b(star fmt(3)) se(par([ ]) fmt(3)))  ///
			legend nolabel starlevels( * 0.10 ** 0.05 *** 0.010) drop(new_offender d_homicide d_assault d_sex_crime d_other_violent d_property d_conspiracy d_other_crime cri_d_drug_any cri_* sec_* dias_interno _W* _X* _cons ed_* minors)     ///
			stats(r2 N, fmt(3 0) label(R-2 Obs.))
		
	
			
			
			
			
